clear all 
set more off 
set maxvar 15000 
clear matrix


	use "$Mydirectory1/3_Output/2_PooledData_analysis.dta", clear 
    keep if baseline_sample==1
    local weight "wgt_sex_race"

    foreach var in ige rank  {
    preserve
    estimates clear 
        
    forval i=1(1)6 {
        gen est_`i'=. 
        gen est_lb_`i' =.
        gen est_ub_`i' =.
    }
    
        foreach fix in 1 2 3 4 5 {

        * Independent variables
            if "`var'"=="ige" & "`fix'"=="1" local indep "log_father_baseline"
            if "`var'"=="ige" & "`fix'"=="2" local indep "log_father_personalinc_interp"
            if "`var'"=="ige" & "`fix'"=="3" local indep "log_father_closest_census_v2"
            if "`var'"=="ige" & "`fix'"=="5" local indep "log_father_1950occscore"
            if "`var'"=="ige" & "`fix'"=="4" local indep "log_father_interp_altwgt"

            if "`var'"=="rank" & "`fix'"=="1" local indep "rank_father_baseline"
            if "`var'"=="rank" & "`fix'"=="2" local indep "rank_father_personalinc"
            if "`var'"=="rank" & "`fix'"=="3" local indep "rank_father_closest_census_v2"
            if "`var'"=="rank" & "`fix'"=="5" local indep "rank_father_1950occscore"
            if "`var'"=="rank" & "`fix'"=="4" local indep "rank_father_byr_bys_altwgt"
            
        * Dependent variables
            if "`var'"=="ige" local dep "log_son_baseline"
            if "`var'"=="rank" local dep "rank_son_baseline"
                    
            
            levelsof decade, local(decades)
            
        * Regressions 
            foreach x of local decades {
            eststo : quietly reg `dep' `indep'  if decade==`x' [pw=`weight'], robust 
                replace est_`fix' = _b[`indep'] if decade==`x'
                replace est_ub_`fix' = _b[`indep']+1.96*_se[`indep'] if decade==`x'
                replace est_lb_`fix' = _b[`indep']-1.96*_se[`indep'] if decade==`x'
                }

            }
        
    * Figures
        bysort decade: keep if _n==1
        keep decade est_*
        
        reshape long est_ est_lb_ est_ub_, i(decade) j(estimate)
        replace decade= decade+1 if estimate==2 
        replace decade= decade+2 if estimate==3
        replace decade= decade+3 if estimate==4
        replace decade= decade+4 if estimate==5
        replace decade= decade+5 if estimate==6
        
    if "`var'"=="ige" {
        local title "IGE"
        local ub "1"
        local lb "0"
        local skip "0.25"
        local maincolor "midblue"
        local color2 "navy"
        local color3 "green"
        local color4 "mint*1.2"
        local color5 "orange*0.5"
    }

    if "`var'"=="rank" {
        local title "Rank"
        local ub "0.45"
        local lb "0.15"
        local skip "0.1"
        local maincolor "orange"
        local color2 "cranberry"
        local color3 "pink*0.75"
        local color4 "lavender"
        local color5 "gray"
    }
        
    if "`var'"=="ige" {
    #delimit ;
    twoway (scatter est_ decade if estimate==1,  msymbol(circle) mcolor(`maincolor') msize(medium) ) 
           (rcap est_lb_  est_ub_  decade if estimate==1, lpatter(solid) lcolor(`maincolor') lwidth(0.5))
            
           (scatter est_ decade if estimate==2,  msymbol(diamond) mcolor(`color2') msize(small) ) 
           (rcap est_lb_  est_ub_  decade if estimate==2, lpatter(solid) lcolor(`color2') lwidth(0.5) )
           
            (scatter est_ decade if estimate==3,  msymbol(triangle) mcolor(`color3') msize(small) ) 
           (rcap est_lb_  est_ub_  decade if estimate==3, lpatter(solid) lcolor(`color3') lwidth(0.5) )
           
            (scatter est_ decade if estimate==4,  msymbol(square) mcolor(`color4') msize(small) ) 
           (rcap est_lb_  est_ub_  decade if estimate==4, lpatter(solid) lcolor(`color4') lwidth(0.5) ) 
           
            (scatter est_ decade if estimate==5,  msymbol(square_hollow) mcolor(`color5') msize(small) ) 
           (rcap est_lb_  est_ub_  decade if estimate==5, lpatter(solid) lcolor(`color5') lwidth(0.5) ) 
           ,
           
    xti(" " "Decade of respondent's birth") xlabel(1910(10)1970) xscale(range(1905 1975))
    ylabel(`lb'(`skip')`ub', axis(1)) yti("`title' coefficient" " ", axis(1)) 
    legend(on ring(0) pos(8) rows(2) order(1 "Baseline" 3 "Father income" 5 "Closest Census" 7 "Alt. inc. score weights" 9 "1950 IPUMS occscore"))
    xlabel(1910 "1910s" 1920 "1920s" 1930 "1930s" 1940 "1940s" 1950 "1950s" 1960 "1960s" 1970 "1970s", labsize(small) ) ;  
    #delimit cr
    graph export "$Mydirectory2/appendix_b/`title'_overtime_incomevariations.pdf", as(pdf) replace
    }
    
    if "`var'"=="rank" {
    #delimit ;
    twoway (scatter est_ decade if estimate==1,  msymbol(circle) mcolor(`maincolor') msize(medium) ) 
           (rcap est_lb_  est_ub_  decade if estimate==1, lpatter(solid) lcolor(`maincolor') lwidth(0.5))
           
           (scatter est_ decade if estimate==2,  msymbol(diamond) mcolor(`color2') msize(small) ) 
           (rcap est_lb_  est_ub_  decade if estimate==2, lpatter(solid) lcolor(`color2') lwidth(0.5))
           
           (scatter est_ decade if estimate==3,  msymbol(triangle) mcolor(`color3') msize(small)) 
           (rcap est_lb_  est_ub_  decade if estimate==3, lpatter(solid) lcolor(`color3') lwidth(0.5))
           
            (scatter est_ decade if estimate==4,  msymbol(square) mcolor(`color4') msize(small)) 
           (rcap est_lb_  est_ub_  decade if estimate==4, lpatter(solid) lcolor(`color4') lwidth(0.5))
           
            (scatter est_ decade if estimate==5,  msymbol(square_hollow) mcolor(`color5') msize(small)) 
           (rcap est_lb_  est_ub_  decade if estimate==5, lpatter(solid) lcolor(`color5') lwidth(0.5))
           
           ,
           
    xti(" " "Decade of respondent's birth") xlabel(1910(10)1970) xscale(range(1905 1975))
    ylabel(`lb'(`skip')`ub', axis(1)) yti("`title' coefficient" " ", axis(1)) 
    legend(on ring(0) pos(2) rows(3) order(1 "Baseline" 3 "Father income" 5 "Closest Census" 7 "Alt. inc. score weights" 9 "1950 IPUMS occscore"))
    xlabel(1910 "1910s" 1920 "1920s" 1930 "1930s" 1940 "1940s" 1950 "1950s" 1960 "1960s" 1970 "1970s", labsize(small) ) ;  
    #delimit cr
    graph export "$Mydirectory2/appendix_b/`title'_overtime_incomevariations.pdf", as(pdf) replace
    }
          
    restore

    }
